Problem:
Score : 400 points

Problem Statement

Along a road running in an east-west direction, there are A shrines and B temples.
The i-th shrine from the west is located at a distance of s_i meters from the west end of the road, and the i-th temple from the west is located at a distance of t_i meters from the west end of the road.

Answer the following Q queries:

Query i (1 \leq i \leq Q): If we start from a point at a distance of x_i meters from the west end of the road and freely travel along the road, what is the minimum distance that needs to be traveled in order to visit one shrine and one temple? (It is allowed to pass by more shrines and temples than required.)

Constraints

1 \leq A, B \leq 10^5

1 \leq Q \leq 10^5

1 \leq s_1 < s_2 < ... < s_A \leq 10^{10}

1 \leq t_1 < t_2 < ... < t_B \leq 10^{10}

1 \leq x_i \leq 10^{10}

s_1, ..., s_A, t_1, ..., t_B, x_1, ..., x_Q are all different.

All values in input are integers.

Input

Input is given from Standard Input in the following format:

A B Q
s_1
:
s_A
t_1
:
t_B
x_1
:
x_Q

Output

Print Q lines. The i-th line should contain the answer to the i-th query.

Sample Input 1

2 3 4
100
600
400
900
1000
150
2000
899
799

Sample Output 1

350
1400
301
399

There are two shrines and three temples. The shrines are located at distances of 100, 600 meters from the west end of the road, and the temples are located at distances of 400, 900, 1000 meters from the west end of the road.

Query 1: If we start from a point at a distance of 150 meters from the west end of the road, the optimal move is first to walk 50 meters west to visit a shrine, then to walk 300 meters east to visit a temple.

Query 2: If we start from a point at a distance of 2000 meters from the west end of the road, the optimal move is first to walk 1000 meters west to visit a temple, then to walk 400 meters west to visit a shrine. We will pass by another temple on the way, but it is fine.

Query 3: If we start from a point at a distance of 899 meters from the west end of the road, the optimal move is first to walk 1 meter east to visit a temple, then to walk 300 meters west to visit a shrine.

Query 4: If we start from a point at a distance of 799 meters from the west end of the road, the optimal move is first to walk 199 meters west to visit a shrine, then to walk 200 meters west to visit a temple.

Sample Input 2

1 1 3
1
10000000000
2
9999999999
5000000000

Sample Output 2

10000000000
10000000000
14999999998

The road is quite long, and we may need to travel a distance that does not fit into a 32-bit integer.

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

import qualified Data.Text as T
import qualified Data.Text.IO as T
import qualified Data.Text.Read as T
import qualified Data.Vector.Unboxed as VU
import Data.Bool (bool)

main :: IO ()
main = do
  [a, b, q] <- map unsafeTextToInt . T.words <$> T.getLine :: IO [Int]
  contents <- unsafeTextToVectorInt <$> T.getContents :: IO (VU.Vector Int)
  let
    ss = VU.slice 0 a contents :: VU.Vector Int
    ts = VU.slice a b contents :: VU.Vector Int
    xs = VU.slice (a + b) q contents :: VU.Vector Int
    dist :: Int -> Int
    dist x = minimum [x - min sL tL, min (x - sL) (tR - x) + (tR - sL), min (x - tL) (sR - x) + (sR - tL), max sR tR - x]
      where
        (sL, sR) = case binaryMaxIndex (< x) ss of
                     Nothing -> (-10 ^ 11, ss VU.! 0)
                     Just i  -> (ss VU.! i, bool (10 ^ 11) (ss VU.! succ i) (i < pred a))
        (tL, tR) = case binaryMaxIndex (< x) ts of
                     Nothing -> (-10 ^ 11, ts VU.! 0)
                     Just i  -> (ts VU.! i, bool (10 ^ 11) (ts VU.! succ i) (i < pred b))
  VU.mapM_ (print . dist) $ xs

binaryMaxIndex :: VU.Unbox a => (a -> Bool) -> VU.Vector a -> Maybe Int
binaryMaxIndex p = binaryMaxInt <$> ((p .) . (VU.!)) <*> (,) 0 . pred . VU.length
{-# INLINE binaryMaxIndex #-}

binaryMaxInt :: (Int -> Bool) -> (Int, Int) -> Maybe Int
binaryMaxInt p = check
  where
    check :: (Int, Int) -> Maybe Int
    check ij@(i, _) | not (p i) = Nothing
                    | otherwise = Just $ unsafeBinaryMaxInt p ij
{-# INLINE binaryMaxInt #-}

unsafeBinaryMaxInt :: (Int -> Bool) -> (Int, Int) -> Int
unsafeBinaryMaxInt p = check
  where
    check :: (Int, Int) -> Int
    check ij@(_, j) | p j       = j
                    | otherwise = search ij
    search :: (Int, Int) -> Int
    search = fst . until ((==) <$> fst <*> pred . snd) halve
      where
        halve :: (Int, Int) -> (Int, Int)
        halve (i, j) | p m       = (m, j)
                     | otherwise = (i, m)
          where
            m = (i + j) `div` 2
{-# INLINE unsafeBinaryMaxInt #-}

unsafeTextToVectorInt :: T.Text -> VU.Vector Int
unsafeTextToVectorInt = VU.unfoldr next
  where
    next :: T.Text -> Maybe (Int, T.Text)
    next s | T.null s  = Nothing
           | otherwise = Just . either undefined (T.stripStart <$>) . T.signed T.decimal $ s
{-# INLINE unsafeTextToVectorInt #-}

unsafeTextToInt :: T.Text -> Int
unsafeTextToInt s = case T.signed T.decimal s of
  Right (n, _) -> n
{-# INLINE unsafeTextToInt #-}